

Istruzioni BASIC per
l'elaborazione di
matrici



Il linguaggio BASIC P6066 offre la possibilità di elaborare le matrici numeriche con le stesse regole del calcolo matriciale. Nel seguente paragrafo sono descritte le istruzioni del linguaggio BASIC che permettono di eseguire le elaborazioni suddette. Per poter eseguire queste istruzioni si deve aver prima eseguito un comando OPTIONS specificando l'operando MAT. Le istruzioni di elaborazione delle matrici sono denominate istruzioni di assegnazione perchè il risultato del calcolo è assegnato ad una matrice specificata; esse sono:

```
MAT matrix = matrix
MAT matrix = matrix + matrix
MAT matrix = matrix - matrix
MAT matrix = (num-exp) * matrix
MAT matrix = matrix * matrix
MAT matrix = CON (num-exp, num-exp)
MAT matrix = IDN (num-exp, num-exp)
MAT matrix = INV (matrix)
MAT matrix = TNR (matrix)
MAT matrix = ZER (num-exp, num-exp)
```

Nelle suddette istruzioni la matrice è considerata un elemento sintattico; infatti il campo specificato con "matrix" deve essere sostituito con il nome di una matrice. Vedremo, caso per caso, se lo stesso nome di matrice può essere espresso a destra ed a sinistra del segno uguale e se la matrice deve avere le due dimensioni uguali (matrice quadrata) o può averle diverse (matrice rettangolare). Per quanto riguarda lo spazio di memoria principale richiesto da una matrice si veda l'appendice F.

Come già visto, per una matrice si distinguono le dimensioni di allocazione da quelle attuali. Le dimensioni di allocazione sono dichiarate esplicitamente con una istruzione DIM e stabiliscono rispettivamente il numero di righe e di colonne riservate per la matrice (in ultima analisi il numero di elementi che la

possono costituire). Se nell'ambito del programma non vi sono istruzioni DIM riferite ad una matrice, s'intende che per essa sono riservate 10 righe e 10 colonne. Le dimensioni attuali sono quelle effettivamente utilizzate dalla matrice, ossia le righe e colonne che realmente utilizza.

Alcune istruzioni non si limitano ad assegnare agli elementi di una matrice dei nuovi valori come risultato della elaborazione definita dal rispettivo algoritmo (ad es. moltiplicazione, righe per colonne, di due matrici etc.), ma definiscono delle nuove dimensioni attuali per la matrice suddetta, il cui prodotto sarà sempre minore o uguale al prodotto delle dimensioni di allocazione. Per quando riguarda le assegnazioni di valori agli elementi di una matrice, si possono verificare le seguenti situazioni a seconda del tipo di precisione dichiarata per la matrice:

1. Nel caso che la matrice sia dichiarata in singola precisione:

- se ad un elemento della matrice viene assegnato un valore in virgola mobile la cui mantissa è al di fuori del campo di rappresentazione, ma il cui esponente rientra nel campo suddetto, il valore numerico viene troncato alle prime 6 cifre significative
- se ad un elemento della matrice è assegnato un valore il cui esponente è al di fuori del campo di rappresentazione, il valore è eguagliato a + 9.99999E63 (se il valore era positivo) od a -9.99999E63 (se il valore era negativo)
- se ad un elemento di una matrice è assegnato un valore compreso nella zona di UNDERFLOW per la singola precisione, il valore è eguagliato a zero

2. Nel caso che la matrice sia dichiarata in doppia precisione (dichiarazione implicita):

- se ad un elemento della matrice è assegnato un valore compreso nella zona di OVERFLOW, il valore è eguagliato a 9.999999999999E99 (se il valore era positivo) oppure a - 9.999999999999E99 (se il valore era negativo)

- se ad un elemento di una matrice è assegnato un valore compreso nella zona di UNDERFLOW, il valore è uguagliato a zero

Quando si verifica ognuno dei casi suddetti (meno il primo caso del punto 1.) il sistema commuta nello stato di debugging dopo aver effettuato l'assegnazione specificata caso per caso. L'utente può far continuare l'esecuzione del programma premendo **CONTINUE** oppure far terminare l'esecuzione stessa premendo il tasto **BREAK**. (Il sistema commuta nello stato comandi). Si noti inoltre che tutte le istruzioni suddette possono far riferimento, come casi particolari, a vettori. Infatti, se una delle due dimensioni di una matrice è dichiarata uguale ad 1, la matrice è in sostanza un vettore. Nell'ultima parte del paragrafo sono descritte le istruzioni che permettono di assegnare agli elementi delle matrici dei valori da tastiera (MAT INPUT), da file dati interno (MAT READ), da file dati esterno (MAT READ:), le istruzioni che permettono di stampare i valori degli elementi delle matrici (MAT PRINT e MAT PRINT USING) e di registrare detti valori su file dati esterno (MAT WRITE:).

Le istruzioni suddette possono avere come operandi non solo delle matrici numeriche, ma anche delle variabili multiple di tipo stringa. Con eccezione dell'istruzione INPUT, le altre istruzioni possono avere più di un operando ed in questo caso si possono avere variabili multiple di diverso tipo, numeriche o stringa, contemporaneamente. Anche le suddette istruzioni permettono (meno l'istruzione MAT WRITE:) di modificare le dimensioni attuali delle variabili multiple a cui si riferiscono. Negli esempi riportati nella descrizione delle istruzioni che segue sono utilizzate le istruzioni MAT INPUT, MAT PRINT e MAT READ, per cui ad una prima lettura del manuale si consiglia di leggere prima le descrizioni relative a tali istruzioni, questo faciliterà la comprensione degli esempi suddetti.



Istruzione MAT ... =

Funzione

Assegna i valori degli elementi di una matrice agli elementi di un'altra matrice.

Formato

MAT matrix = matrix

dove:

matrix

indica il nome di una matrice numerica quadrata o rettangolare.

Azione

Ogni valore di ogni elemento della matrice specificata sulla destra del segno uguale è assegnato al corrispondente elemento della matrice alla sinistra del segno uguale.

La matrice alla sinistra del segno uguale assume le dimensioni attuali della matrice a destra del segno uguale.

Nota

Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto delle dimensioni attuali della matrice a destra del segno uguale.

Esempi

1. La routine sottostante richiede (istruzione 20) nove dati numerici da tastiera da assegnare agli elementi della matrice A (avente dimensioni di allocazione di 10 x 10 elementi) che viene ad assumere dimensioni attuali pari a 3 x 3 elementi. L'istruzione 30 assegna quindi i valori degli elementi della matrice A agli elementi della matrice B che assume dimensioni attuali di 3 x 3 elementi. Dalla stampa prodotta in seguito all'esecuzione della routine si

può vedere il risultato ottenuto.

```
LIST
FILE

0010 DISP "Introduci i valori
0020 MAT INPUT A(3,3)
0030 MAT B=A
0040 MAT PRINT A;
0050 PRINT
0060 MAT PRINT B;
0070 END

END OF LISTING

RUN
Introduci i valori
1,2,3,4,5,6,7,8,9
 1 2 3
 4 5 6
 7 8 9

 1 2 3
 4 5 6
 7 8 9
```

2. La routine sottostante dichiara, istruzione 10, per la matrice A e per la matrice B dimensioni di allocazione rispettivamente di 3×3 e 2×3 elementi. L'istruzione 30 richiede 9 dati numerici da tastiera da assegnare alla matrice A, mentre l'istruzione 70 assegna agli elementi della matrice B i valori numerici specificati nella istruzione 140. Infine l'istruzione 80 assegna agli elementi della matrice A i valori degli elementi della matrice B e la matrice A assume le dimensioni attuali di 2×3 elementi. Dalla stampa prodotta con l'esecuzione del programma si può vedere il risultato ottenuto.

```
LIST
FILE *MAT2

0010 DIM A(3,3),B(2,3)
0020 DISP "INTRODUCI I VALORI PER A(3,3)";
0030 MAT INPUT A
0040 PRINT "LA MATRICE A HA I VALORI:"
0050 MAT PRINT A;
0060 PRINT
0070 MAT READ B
0080 MAT A=B
0090 PRINT "LA MATRICE B HA I VALORI:"
0100 MAT PRINT B;
0110 PRINT
```

```
0120 PRINT "ORA LA MATRICE A HA I VALORI:"
0130 MAT PRINT A:
0140 DATA 1,2,3,4,5,6
0150 END
```

END OF LISTING

RUN
**** FORMALLY CORRECT PROGRAM ****

INTRODUCI I VALORI PER A0

9,9,9,9,9,9,9,9,9

LA MATRICE A HA I VALORI:

```
9 9 9
9 9 9
9 9 9
```

LA MATRICE B HA I VALORI:

```
1 2 3
4 5 6
```

ORA LA MATRICE A HA I VALORI:

```
1 2 3
4 5 6
```


Istruzione MAT ... +

Funzione

Esegue l'operazione di addizione tra due matrici e ne assegna il risultato alla matrice specificata prima del segno uguale.

Formato

MAT matrix = matrix + matrix

dove:

matrix

indica il nome di una matrice numerica quadrata rettangolare.

Azione

I valori degli elementi corrispondenti delle due matrici numeriche a destra del segno uguale sono addizionati ed i risultati della operazione sono assegnati ai corrispondenti elementi della matrice specificata a sinistra del segno uguale.

La matrice a sinistra del segno uguale assume le dimensioni attuali delle matrici a destra del segno uguale.

Note

1. La somma tra i valori degli elementi delle matrici a destra del segno uguale è una somma algebrica.
2. Le due matrici a destra del segno uguale devono avere le stesse dimensioni attuali.
3. Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto delle dimensioni attuali delle due matrici a destra del segno uguale.
4. La matrice che compare a sinistra del segno uguale può comparire anche a destra del segno uguale una

o due volte.

Esempi

1. La seguente routine calcola la somma tra due matrici B e C ed assegna il risultato ad una terza matrice A.

```
LIST
FILE      *MAT4

0010 DIM A(3,2),B(3,2),C(3,2)
0020 DISP "Introduci i valori per B      ";
0030 MAT INPUT B
0040 DISP "Introduci i valori per C      ";
0050 MAT INPUT C
0060 MAT A=B+C
0070 PRINT "I valori della matrice B sono:"
0080 MAT PRINT B;
0090 PRINT
0100 PRINT "I valori della matrice C sono:"
0110 MAT PRINT C;
0120 PRINT "I valori della matrice A sono:"
0130 MAT PRINT A;
0140 END
```

END OF LISTING

```
RUN
Introduci i valori per B      ?
1,2,3,4,5,6
Introduci i valori per C      ?
1,2,3,4,5,6
I valori della matrice B sono:
 1  2
 3  4
 5  6

I valori della matrice C sono:
 1  2
 3  4
 5  6
I valori della matrice A sono:
 2  4
 6  8
10 12

RUN
Introduci i valori per B      ?
-25,55,15,-12,+7,-10,150
Introduci i valori per C      ?
5,12,-7,-12,+11,55,+15,8,-150
I valori della matrice B sono:
 5,12 -7
-12 11,55
 15,8 -150

I valori della matrice C sono:
-25,55 15
-12 7
-10 150
I valori della matrice A sono:
-20,43 8
-24 18,55
 5,8 8
```

2. Nell'esempio seguente si può vedere come non sia possibile eseguire l'addizione di due matrici che non abbiano lo stesso numero di righe e lo stesso numero di colonne, come dimensioni attuali. L'esecuzione della istruzione 60, infatti, produce una segnalazione di errore, come si può vedere. Premendo **BREAK** il sistema commuta nello stato di comandi. Con il comando FETCH (FET 10) si richiama nel buffer di tastiera l'istruzione 10 e la si modifica ponendo B (3,2) invece di B (2,3), per cui una successiva esecuzione (comando RUN) produce la stampa dei valori di B (istruzione 80) di C (istruzione 110) e di A (istruzione 130).

```
LIST
FILE      *MAT5
```

```
0010 DIM A(4,4),B(3,2),C(2,3)
0020 DISP "Introduci i valori per B      ";
0030 MAT INPUT B
0040 DISP "Introduci i valori per C      ";
0050 MAT INPUT C
0060 MAT A=B+C
0070 PRINT "I valori della matrice B sono:"
0080 MAT PRINT B;
0090 PRINT
0100 PRINT "I valori della matrice C sono:"
0110 MAT PRINT C;
0120 PRINT "I valori della matrice A sono:"
0130 MAT PRINT A;
0140 END
```

```
END OF LISTING
```

```
RUN
Introduci i valori per B      ?
1,2,3,4,5,6
Introduci i valori per C      ?
1,2,3,4,5,6
ERROR 73 IN LINE 60
FET 10
0010 DIM A(4,4),B(3,2),C(2,3)
0010 DIM A(4,4),B(3,2),C(3,2)
RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per B      ?
1,2,3,4,5,6
Introduci i valori per C      ?
4,4,4,4,4,4
I valori della matrice B sono:
 1 2
 3 4
 5 6

I valori della matrice C sono:
 4 4
 4 4
 4 4
I valori della matrice A sono:
 5 6
 7 8
 9 10
```

3. In questo esempio si dichiara la matrice A (istruzione 5) in singola precisione. Al primo elemento della matrice B si assegna da tastiera un valore con esponente nella zona di OVERFLOW per la rappresentazione in singola precisione (10E78), istruzione 30, mentre all'ultimo elemento della matrice C si assegna un valore nella zona di UNDERFLOW per la rappresentazione in singola precisione (1E-85). Quando viene eseguita l'istruzione 60 il sistema assegna al primo elemento della matrice A il valore 9.999999E63 ed all'ultimo elemento della stessa matrice il valore 6 (a 6 viene sommato lo zero) come si può vedere dalla stampa prodotta con l'istruzione 130. Viene visualizzato l'errore sottoriportato ed il sistema è nello stato di debugging. Premendo il tasto **CONTINUE** l'esecuzione prosegue e vengono prodotte le stampe sottoriportate.

```

LIST
FILE      *MAT6

0005 DCL S(A(5))
0010 DIM A(4,4),B(3,2),C(3,2)
0020 DISP "Introduci i valori per B          ";
0030 MAT INPUT B
0040 DISP "Introduci i valori per C          ";
0050 MAT INPUT C
0060 MAT A=B+C
0070 PRINT "I valori della matrice B sono:"
0080 MAT PRINT B;
0090 PRINT
0100 PRINT "I valori della matrice C sono:"
0110 MAT PRINT C;
0120 PRINT "I valori della matrice A sono:"
0130 MAT PRINT A;
0140 END

END OF LISTING

RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per B          ?
10E78,2,3,4,5,6
Introduci i valori per C          ?
1,2,3,4,5,1E-85
ERROR 3  IN LINE 60
I valori della matrice B sono:
 1.0000000E+79  2
 3  4
 5  6

I valori della matrice C sono:
 1  2
 3  4
 5  1.0000000E-85
I valori della matrice A sono:
9.9999900E+63  4
 6  8
10  6

```

Istruzione MAT ... -

Funzione

Esegue l'operazione di sottrazione tra due matrici e ne assegna il risultato ad una matrice specificata.

Formato

MAT matrix = matrix - matrix

dove:

matrix

indica il nome di una matrice numerica quadrata o rettangolare.

Azione

Ai valori degli elementi della prima matrice a destra del segno uguale sono sottratti i valori degli elementi corrispondenti della seconda matrice a destra del segno uguale; i risultati sono assegnati ai corrispondenti elementi della matrice specificata a sinistra del segno uguale.

La matrice a sinistra del segno uguale assume le dimensioni attuali delle matrici a destra del segno uguale.

Note

1. La differenza tra i valori degli elementi delle matrici a destra del segno uguale è una differenza algebrica.
2. Le due matrici a destra del segno uguale devono avere le stesse dimensioni attuali.
3. Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto delle dimensioni attuali delle due matrici a destra del segno uguale.
4. La matrice che compare a sinistra del segno uguale

può comparire anche a destra del segno uguale una o due volte.

Esempio

La routine sottostante calcola la differenza algebrica tra due matrici B e C.

```
LIST
FILE      *MAT7
```

```
0010 DIM A(25,35),B(5,5),C(5,5)
0020 DISP "Introduci i valori per B      ";
0030 MAT INPUT B
0040 DISP "Introduci i valori per C      ";
0050 MAT INPUT C
0060 MAT A=B-C
0070 PRINT
0080 PRINT "I valori di B sono:"
0090 MAT PRINT B;
0100 PRINT
0110 PRINT "I valori di C sono:"
0120 MAT PRINT C;
0130 PRINT
0140 PRINT "I valori di A sono:"
0150 MAT PRINT A;
0160 END
```

```
END OF LISTING
```

```
RUN
```

```
Introduci i valori per B      ?
1,2,3,6,5,4,7,9,8,1,2,6,3,5,8,9,7,4,1,2,-6,-9,-8,-9,-5
Introduci i valori per C      ?
-9,-5,-8,-9,5,-2,-6,5,8,9,5,5,2,6,6,9,3,5,8,-9,-5,-6,-9,-3,-5
```

```
I valori di B sono:
```

```
 1  2  3  6  5
 4  7  9  8  1
 2  6  3  5  8
 9  7  4  1  2
-6 -9 -8 -9 -5
```

```
I valori di C sono:
```

```
-9 -5 -8 -9  5
-2 -6  5  8  9
 5  5  2  6  6
 9  3  5  8 -9
-5 -6 -9 -3 -5
```

```
I valori di A sono
```

```
10  7 11 15  0
 6 13  4  0 -8
-3  1  1 -1  2
 0  4 -1 -7 11
-1 -3  1 -6  0
```

Istruzione
MAT...* Scalare

Funzione

Moltiplica ogni elemento di una matrice per il valore di una espressione numerica e ne assegna il risultato ad un'altra matrice specificata.

Formato

MAT matrix = (num-exp) * matrix

dove:

num-exp

indica una espressione numerica che viene moltiplicata per ogni valore degli elementi della matrice specificata a destra del segno uguale

matrix

indica il nome di una matrice numerica quadrata o rettangolare.

Azione

L'espressione numerica è eseguita ed il valore ottenuto è moltiplicato per il valore di ogni elemento della matrice numerica a destra del segno uguale. Il risultato ottenuto è assegnato al corrispondente elemento della matrice a sinistra del segno uguale.

La matrice a sinistra del segno uguale assume le dimensioni attuali della matrice a destra del segno uguale.

Note

1. La matrice a sinistra del segno uguale non deve avere il prodotto delle dimensioni di allocazione minore del prodotto delle dimensioni attuali della matrice a destra del segno uguale.
2. La matrice che compare a sinistra del segno uguale può comparire anche a destra del segno uguale.

Esempio

La routine seguente esegue il prodotto scalare di una variabile numerica con una matrice, istruzione 40 e di una costante numerica con una matrice, istruzione 130.

```
LIST
FILE      *MAT8

0010 MAT READ A(3,4)
0020 DISP "Introduci moltiplicatore di A      ";
0030 INPUT B
0040 MAT C=(B)*A
0050 PRINT
0060 PRINT "Il valore di B e'";B
0070 PRINT
0080 PRINT "I valori di A sono:"
0090 MAT PRINT A;
0100 PRINT
0110 PRINT "I valori di C sono:"
0120 MAT PRINT C;
0130 MAT C=(5)*C
0140 PRINT
0150 PRINT "Ora i valori di C sono:"
0160 MAT PRINT C;
0170 DATA 1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16
0180 END
```

END OF LISTING

```
RUN
Introduci moltiplicatore di A      ?
-10
```

Il valore di B e' -10

I valori di A sono:
1 2 3 4
5 6 7 8
9 10 11 12

I valori di C sono:
-10 -20 -30 -40
-50 -60 -70 -80
-90 -100 -110 -120

Ora i valori di C sono:
-50 -100 -150 -200
-250 -300 -350 -400
-450 -500 -550 -600

Istruzione MAT ... *

Funzione

Esegue il prodotto, righe per colonne, tra due matrici e ne assegna il risultato ad una matrice specificata.

Formato

MAT matrix = matrix * matrix

dove:

matrix

indica il nome di una matrice numerica quadrata o rettangolare.

Azione

Ogni elemento di una riga (i) della prima matrice numerica a destra del segno uguale è moltiplicato per l'elemento che compare nello stesso ordine in una colonna (j) della seconda matrice numerica a destra del segno uguale.

I prodotti così ottenuti sono sommati; il risultato ottenuto è assegnato all'elemento della matrice a sinistra del segno uguale che si trova nella riga i e nella colonna j.

Se si moltiplica una matrice A di dimensioni attuali (p,m) con una matrice B di dimensioni attuali (m,n) si ottiene una matrice C di dimensioni attuali (p,n) tale che per $i = 1, 2, \dots, p$ e per $j = 1, 2, \dots, n$:

$$c_{i,j} = \sum_{k=1}^m a_{i,k} * b_{k,j}$$

dove $c_{i,j}$ è un elemento generico della matrice C e $a_{i,k}$ e $b_{k,j}$ sono elementi rispettivamente della matrice A e B.

Note

1. La matrice numerica che compare a sinistra del segno uguale non deve comparire anche a destra del segno uguale.
2. Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto del numero attuale di righe della prima matrice a destra del segno uguale per il numero attuale di colonne della seconda matrice a destra del segno uguale.
3. Il numero di colonne della prima matrice a destra del segno uguale deve essere uguale al numero di righe della seconda matrice a destra del segno uguale.

Esempio

La routine sottostante calcola il prodotto, righe per colonne, della matrice A con la matrice B. Con le stampe prodotte durante l'esecuzione della routine si possono vedere i valori assegnati alla matrice A, alla matrice B, ed alla matrice C.

```
LIST
FILE      *MAT9

0010 REM Ecco il prodotto righe per colonne tra la matrice A e la matrice B
0020 DIM A(2,3),B(3,2)
0030 DISP "Introduci i valori per A      ";
0040 MAT INPUT A
0050 DISP "Introduci i valori per B      ";
0060 MAT INPUT B
0070 MAT C=A*B
0080 PRINT
0090 PRINT "I valori di A sono:"
0100 MAT PRINT A;
0110 PRINT
0120 PRINT "I valori di B sono:"
0130 MAT PRINT B;
0140 PRINT
0160 PRINT "I valori di C sono:"
0170 MAT PRINT C;
0180 END

END OF LISTING
```

```
RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per A      ?
1,-5,+8,-3,4,-8
Introduci i valori per B      ?
0,-5,-4,4,8,+6

I valori di A sono:
 1 -5 8
-3 4 -8

I valori di B sono:
 0 -5
-4 4
 8 6

I valori di C sono:
 84 23
-80 -17
```




Istruzione MAT ...CON

Funzione

Assegna il valore uno ad ogni elemento di una matrice.

Formato

MAT matrix = CON [(num-exp, num-exp)]

dove:

matrix

indica il nome di una matrice numerica quadrata o rettangolare

num-exp

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta la nuova dimensione attuale della matrice specificata a sinistra del segno uguale.

Azione

A tutti gli elementi della matrice numerica a sinistra del segno uguale è assegnato il valore 1.

Se è presente la parte opzionale, la matrice numerica a sinistra del segno uguale assume le dimensioni attuali pari ai valori ottenuti eseguendo le due espressioni numeriche indicate tra parentesi (che sono arrotondate all'intero più prossimo). Se la parte opzionale non è specificata le dimensioni attuali della matrice non sono modificate.

Nota

Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto dei valori, arrotondati all'intero più prossimo, ottenuti eseguendo le due espressioni numeriche della parte opzionale.

Esempio

La routine sottostante pone in evidenza come l'istruzione che assegna la costante uno agli elementi di una matrice ne permette anche il ridimensionamento. L'istruzione 70 assegna la costante uno a tutti gli elementi allocati in memoria principale per la matrice Z, come si può vedere con la stampa prodotta dall'istruzione 100. L'istruzione 110 assegna a sei elementi delle prime cinque righe della matrice B la costante uno, come si può vedere dalla stampa prodotta con l'istruzione 140.

```
LIST
FILE      *MAT10

0010 DIM Z(5,5)
0020 DISP "Introduci i valori per B      ";
0030 MAT INPUT B(1,5)
0040 PRINT
0050 PRINT "I valori della matrice B sono:"
0060 MAT PRINT B;
0070 MAT Z=CON
0080 PRINT
0090 PRINT "I valori della matrice Z sono:"
0100 MAT PRINT Z;
0110 MAT B=CON(5,6)
0120 PRINT
0130 PRINT "Ora i valori della matrice B sono:"
0140 MAT PRINT B;
0150 END
```

END OF LISTING

```
RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per B      ?
1,2,3,4,5

I valori della matrice B sono:
 1  2  3  4  5

I valori della matrice Z sono:
 1  1  1  1  1
 1  1  1  1  1
 1  1  1  1  1
 1  1  1  1  1
 1  1  1  1  1

Ora i valori della matrice B sono:
 1  1  1  1  1  1
 1  1  1  1  1  1
 1  1  1  1  1  1
 1  1  1  1  1  1
 1  1  1  1  1  1
```

Istruzione MAT ... IDN

Funzione

Assegna il valore uno a tutti gli elementi della diagonale principale di una matrice quadrata ed il valore zero a tutti gli altri elementi della matrice.

Formato

MAT matrix = IDN [(num-exp, num-exp)]

dove:

matrix

indica il nome di una matrice numerica quadrata
num-exp

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta la nuova dimensione attuale della matrice specificata a sinistra del segno uguale.

Azione

Agli elementi con indici uguali, della matrice numerica quadrata indicata a sinistra del segno uguale, è assegnato il valore 1.

A tutti gli altri elementi della matrice suddetta è assegnato il valore \emptyset .

Se è presente la parte opzionale le espressioni numeriche sono eseguite ed i valori ottenuti sono arrotondati all'intero più prossimo. La matrice numerica quadrata indicata nella istruzione assume le dimensioni attuali specificate dai valori delle espressioni numeriche racchiuse tra parentesi.

Se la parte opzionale non è specificata le dimensioni attuali della matrice specificata non sono modificate.

Note

1. I valori ottenuti eseguendo le due espressioni numeriche indicate nella parte opzionale, arrotondati

all'intero più prossimo, devono essere uguali tra loro e maggiori di zero.

2. Il prodotto delle dimensioni di allocazione della matrice numerica quadrata deve essere maggiore od uguale al prodotto dei valori ottenuti dalle espressioni numeriche specificate tra parentesi.

Esempio

La routine sottostante mostra come anche l'istruzione che assegna ad una matrice i valori della matrice identità permette di ridimensionare le dimensioni attuali. L'istruzione 10 assegna alla matrice A la matrice identità secondo le dimensioni di allocazione (10 x 10). L'istruzione 50 assegna alla stessa matrice A la matrice identità, ma secondo le nuove dimensioni attuali (4,4).

```
LIST
FILE      +MAT11

0010 MAT A=IDN
0020 PRINT
0030 PRINT "I valori di A sono:"
0040 MAT PRINT A;
0050 MAT A=IDN(4,4)
0060 PRINT
0070 PRINT "Ora i valori di A sono:"
0080 MAT PRINT A;
0090 END
```

END OF LISTING.

```
RUN
**** FORMALLY CORRECT PROGRAM ****
```

```
I valori di A sono:
 1 0 0 0 0 0 0 0 0 0
 0 1 0 0 0 0 0 0 0 0
 0 0 1 0 0 0 0 0 0 0
 0 0 0 1 0 0 0 0 0 0
 0 0 0 0 1 0 0 0 0 0
 0 0 0 0 0 1 0 0 0 0
 0 0 0 0 0 0 1 0 0 0
 0 0 0 0 0 0 0 1 0 0
 0 0 0 0 0 0 0 0 1 0
 0 0 0 0 0 0 0 0 0 1
```

```
Ora i valori di A sono:
 1 0 0 0
 0 1 0 0
 0 0 1 0
 0 0 0 1
```

Istruzione MAT ... INV

Funzione Calcola la matrice inversa di una matrice quadrata e la assegna ad una matrice specificata.

Formato **MAT matrix = INV (matrix)**

dove:

matrix

indica il nome di una matrice numerica.

Azione La matrice inversa della matrice numerica quadrata indicata alla destra del segno uguale viene assegnata, elemento per elemento, alla matrice numerica indicata alla sinistra del segno uguale.

Data una matrice numerica quadrata M di dimensioni (m,m) la matrice inversa N, se esiste, è la matrice di eguali dimensioni tali che:

$$M * N = N * M = I$$

dove I è una matrice identità.

Non tutte le matrici hanno una matrice inversa, infatti le matrici con determinante uguale a zero non hanno una matrice inversa.

La matrice a sinistra del segno uguale assume le stesse dimensioni attuali della matrice a destra del segno uguale.

Note

1. Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto delle dimensioni attuali della matrice a destra del segno uguale.

2. Il calcolo della inversione di una matrice fornisce anche il valore del suo determinante, che è fornito al programma utilizzando la funzione di sistema DET. Quindi, se in un programma compare l'istruzione `70 MAT B = INV (A)` ed in una istruzione successiva la funzione DET, quest'ultima ritorna il valore del determinante della matrice A.
3. La matrice a destra del segno uguale deve essere una matrice quadrata.
4. La matrice a sinistra del segno uguale può essere specificata anche a destra del segno uguale.
5. Se il determinante della matrice a destra del segno uguale è zero, è visualizzato un errore di tipo recuperabile ed il sistema è nello stato di debugging; premendo **CONTINUE** si può continuare l'esecuzione del programma.

Esempio

Il seguente programma calcola l'inversa della matrice A e della matrice B, stampa quindi i valori ottenuti per le due matrici ed il valore dei rispettivi determinanti. Sono riportate due esecuzioni del programma. Durante la prima esecuzione i valori forniti da tastiera sono tali da rendere uguale a zero il determinante della matrice A ed il determinante della matrice B, per cui l'esecuzione del programma è interrotta due volte ed è emesso un messaggio di errore recuperabile. Premendo il tasto **CONTINUE** l'esecuzione riprende ogni volta e la seconda volta continua fino al termine.

```

LIST
FILE    +MAT12

0010 DISP "Introduci i valori per A      ";
0020 MAT INPUT A(5,5)
0030 PRINT
0040 PRINT "I valori di A sono:"
0050 MAT PRINT A;
0060 PRINT
0070 MAT B=INVA)
0080 PRINT "I valori di B sono"
0090 MAT PRINT B;
0100 PRINT
0110 PRINT "Il DETERMINANTE di A e' ";DET
0120 MAT B=INVB)
0130 PRINT
0140 PRINT "Il DETERMINANTE di B e' ";DET
0150 PRINT
0160 PRINT "Ora il valore di B e' "

```


Istruzione MAT ... TRN

Funzione

Assegna ad una matrice specificata gli elementi di un'altra matrice scambiando tra loro le righe con le colonne.

Formato

MAT matrix = TRN (matrix)

dove:

matrix

indica il nome di una matrice numerica quadrata e rettangolare.

Azione

Le righe e le colonne della matrice a destra del segno uguale sono scambiate tra loro e la nuova matrice così costruita è assegnata alla matrice indicata a sinistra del segno uguale.

I valori della colonna x della matrice a destra del segno uguale coincidono con i valori della riga x della matrice a sinistra del segno uguale. I valori della riga y della matrice a destra del segno uguale coincidono con i valori della colonna y della matrice a sinistra del segno uguale.

Se la matrice a destra del segno uguale ha dimensioni attuali (m,n) , la matrice a sinistra del segno uguale assume dimensioni attuali (n,m) .

Note

1. Il prodotto delle dimensioni di allocazione della matrice a sinistra del segno uguale deve essere maggiore od uguale al prodotto delle dimensioni attuali della matrice a destra del segno uguale.
2. Non si deve avere la stessa matrice da entrambi i lati del segno uguale.

Esempio

La routine sottostante assegna alla matrice A i valori della matrice B scambiando tra loro le righe con le colonne.

```
LIST
FILE      +MAT13

0010 DISP "Introduci i valori per B      ";
0020 MAT INPUT B(4,3)
0030 MAT A=TRN(B)
0040 PRINT
0050 PRINT "I valori di B sono:"
0060 MAT PRINT B;
0070 PRINT
0080 PRINT "I valori di A sono:"
0090 MAT PRINT A;
0100 END

END OF LISTING

RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per B      ?
10,11,12,21,22,23,31,32,33,41,42,43

I valori di B sono:
10 11 12
21 22 23
31 32 33
41 42 43

I valori di A sono:
10 21 31 41
11 22 32 42
12 23 33 43
```

Istruzione MAT ... ZER

Funzione Assegna il valore zero a tutti gli elementi di una matrice.

Formato **MAT matrix = ZER [(num-exp, num-exp)]**

dove:

matrix

indica il nome di una matrice numerica quadrata o rettangolare

num-exp

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta la nuova dimensione attuale della matrice specificata a sinistra del segno uguale.

Azione A tutti gli elementi della matrice numerica indicata nella istruzione è assegnato il valore zero.

Se è indicata la parte opzionale, le espressioni numeriche sono eseguite ed i valori ottenuti m ed n arrotondati all'intero più prossimo costituiscono le nuove dimensioni attuali della matrice specificata.

Se la parte opzionale non è specificata le dimensioni attuali della matrice specificata non sono modificate.

Nota Il prodotto delle dimensioni di allocazione della matrice specificata nell'istruzione deve essere maggiore od uguale al prodotto $m*n$.

Esempio Il seguente programma assegna il valore zero agli elementi delle prime tre righe e quattro colonne della matrice A ed agli elementi delle prime cinque righe e

cinque colonne della matrice B. Si noti che l'istruzione MAT INPUT assegna alla matrice A dimensioni attuali 3*4 e che l'istruzione 70 assegna alla matrice B dimensioni attuali 5*5.

```
LIST
FILE
```

```
0010 DISP "INTRODUCI I VALORI PER A          ";
0020 MAT INPUT A(3,4)
0030 PRINT
0040 PRINT "I VALORI DI A SONO:"
0050 MAT PRINT A;
0060 MAT A=ZER
0070 MAT B=ZER(5,5)
0080 PRINT
0090 PRINT "ORA I VALORI DI A SONO:"
0100 PRINT
0110 MAT PRINT A;
0120 PRINT
0130 PRINT
0140 PRINT "I VALORI DI B SONO:"
0150 MAT PRINT B;
0160 END
```

```
END OF LISTING
```

```
RUN
```

```
**** FORMALLY CORRECT PROGRAM ****
INTRODUCI I VALORI PER A          ?
1,2,3,4,5,6,7,8,9,0,9,8
```

```
I VALORI DI A SONO:
```

```
1 2 3 4
5 6 7 8
9 0 9 8
```

```
ORA I VALORI DI A SONO:
```

```
0 0 0 0
0 0 0 0
0 0 0 0
```

```
I VALORI DI B SONO:
```

```
0 0 0 0 0
0 0 0 0 0
0 0 0 0 0
0 0 0 0 0
0 0 0 0 0
```

Istruzione MAT INPUT

Funzione Assegna agli elementi di una variabile multipla i dati introdotti da tastiera.

Formato **MAT INPUT array [(num-exp, num-exp)]**

dove:

array

indica il nome di una variabile multipla, numerica o stringa, alla quale sono assegnati i valori introdotti da tastiera

num-exp

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta la nuova dimensione attuale della variabile multipla specificata con array.

Azione

L'esecuzione del programma è interrotta. Sul display un punto interrogativo (?) avverte l'utente di digitare $n*m$ dati, dove m ed n sono i valori corrispondenti alle due espressioni specificate tra parentesi oppure le dimensioni attuali della matrice. I dati digitati sono assegnati nell'ordine, riga per riga, alla variabile multipla indicata nell'istruzione.

Note

1. Ogni dato digitato deve essere separato dai successivi mediante una virgola.
2. Ogni introduzione di dati deve essere completata con la pressione del tasto EOL. Se non sono stati digitati un numero di dati sufficienti ad esaurire la richiesta sul display due punti interrogativi in dicano che il sistema è in attesa di altri dati da tastiera.

3. Quando il numero di dati digitati è pari al numero di quelli richiesti dal sistema, l'esecuzione del programma riprende.
4. Se sono digitati più valori di quelli richiesti dall'istruzione, il sistema visualizza il messaggio TOO MUCH DATA - EXCESS IGNORED e riprende l'esecuzione del programma.
5. Se è presente la parte opzionale, la matrice specificata nella istruzione assume come dimensioni attuali i valori corrispondenti alle due espressioni numeriche.
6. E' utile far precedere l'istruzione MAT INPUT da una istruzione DISP o PRINT che specifichino all'utente quali dati deve introdurre.
7. Ogni dato introdotto da tastiera deve essere dello stesso tipo (numerico o stringa) della variabile multipla a cui il valore è assegnato (stringa o numerica).
8. Se si introducono da tastiera stringhe con spazi iniziali e finali e/o virgole, le stringhe devono essere comprese tra virgolette.
9. Il prodotto delle dimensioni di allocazione della variabile multipla specificata nell'istruzione deve essere maggiore od uguale al prodotto dei valori delle due espressioni numeriche che definiscono le nuove dimensioni attuali.
10. Per ulteriori informazioni si vedano le note relative all'istruzione INPUT.

Esempi

1. La routine sottostante mostra l'impiego della istruzione MAT INPUT. Con l'istruzione 20 si assegnano agli elementi allocati in memoria principale per la matrice A (10 x 10) i valori numerici introdotti da tastiera. Si noti come, dopo ogni introduzione che non esaurisce la richiesta di dati da parte dell'istruzione, è visualizzato un doppio punto interrogativo (??). Con l'istruzione 70 si assegnano ai primi tre elementi delle prime due righe della variabile multipla stringa A\$ le stringhe di caratteri *OLIVETTI P6066*.

```
LIST
FILE    +MAT15
```

```
0010 DISP "Introduci i valori per A      ";
0020 MAT INPUT A
0030 PRINT
0040 PRINT "I valori di A sono:"
0050 MAT PRINT A;
0060 DISP "Introduci i valori per A$     ";
0070 MAT INPUT A$(2,3)
0080 PRINT
0090 PRINT "I valori di A$ sono:"
0100 MAT PRINT A$
0110 END
```

END OF LISTING

RUN

```
Introduci i valori per A      ?
0,1,2,3,4,5,6,7,8,9,0,1,2,3,4,5,6,7,8,9,0,1,2,3,4,5,6,7,8,9
Introduci i valori per A      ??
0,1,2,3,4,5,6,7,8,9,0,1,2,3,4,5,6,7,8,9,0,1,2,3,4,5,6,7,8,9
Introduci i valori per A      ??
0,1,2,3,4,5,6,7,8,9,0,1,2,3,4,5,6,7,8,9
```

I valori di A sono:

```
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
0 1 2 3 4 5 6 7 8 9
```

Introduci i valori per A\$?

```
*Olivetti P6066*,*Olivetti P6066*,*Olivetti P6066*,*Olivetti P6066*
```

Introduci i valori per A\$??

```
*Olivetti P6066*,*Olivetti P6066*
```

I valori di A\$ sono:

```
*Olivetti P6066*
*Olivetti P6066*
```

```
*Olivetti P6066*
*Olivetti P6066*
```

```
*Olivetti P6066*
*Olivetti P6066*
```

2. La routine sottostante dichiara, istruzione 10, per gli elementi della variabile multipla stringa A\$ una dimensione di allocazione di 30 caratteri. Con l'istruzione 30 vengono assegnati ai primi due elementi della prima riga di A\$ le due stringhe di caratteri successivamente stampate con l'istruzione 60.

```
LIST
FILE    +MAT16

0010 DCL 30 A$(
0020 DISP "Introduci i valori per A$          "
0030 MAT INPUT A$(1,2)
0040 PRINT
0050 PRINT "I valori di A$ sono:"
0060 MAT PRINT A$
0070 END

END OF LISTING

RUN
Introduci i valori per A$          ?
Il "Giorno" e` un quotidiano,!!!!!!"#####$$$$$$

I valori di A$ sono:
Il "Giorno" e` un quotidiano  !!!!!!!"#####$$$$$$
```


Se una variabile multipla è seguita da "," il contenuto degli elementi di ogni riga della variabile multipla è stampato partendo dall'inizio di una delle 5 zone di stampa in cui è diviso il tabulato, come indicato in figura 5-2 (vedi istruzione PRINT).

Se una variabile multipla è seguita da ";" il contenuto degli elementi di ogni riga della variabile multipla è stampato di seguito al precedente nell'ambito della stessa linea di stampa.

Se dopo l'ultima variabile multipla di una istruzione MAT PRINT non vi è nè "," nè ";" la stampa del contenuto dei suoi elementi avviene come nel caso in cui vi sia ",".

Esempi

1. Nel programma sottostante l'istruzione 110 stampa i valori delle matrici A, B e C come si vede dopo l'intestazione "I valori di A, B e C, stampati usando ";" come separatore sono: ". Le istruzioni 160 e 170 producono lo stesso risultato. L'istruzione 220 stampa gli stessi valori ma secondo un altro formato, come si può vedere dopo la terza intestazione. Infine le istruzioni 280 e 290 producono lo stesso risultato.

LIST
FILE

```
0010 DISP "INTRODUCI I VALORI PER A      ";
0020 MAT INPUT A(4,4)
0030 DISP "INTRODUCI I VALORI DI B      ";
0040 MAT INPUT B(3,3)
0050 DISP "INTRODUCI I VALORI DI C      ";
0060 MAT INPUT C(5,4)
0070 PRINT
0080 PRINT
0090 PRINT
0100 PRINT "I VALORI DI A,B e C, STAMPATI USANDO ; COME SEPARATORE SONO:"
0110 MAT PRINT A;B;C
0120 PRINT
0130 PRINT
0140 PRINT
0150 PRINT "CON DUE ISTRUZIONI PRODUCO LA STAMPA DI PRIMA."
0160 MAT PRINT A;
0170 MAT PRINT B;C
0180 PRINT
0190 PRINT
0200 PRINT
0210 PRINT "STAMPO I VALORI DI A,B e C USANDO , COME SEPARATORE."
0220 MAT PRINT A,B,C
0230 PRINT
0240 PRINT
0250 PRINT
0260 PRINT "CON DUE ISTRUZIONI PRODUCO LA STAMPA DI PRIMA."
0270 MAT PRINT A,
0280 MAT PRINT B,C
0290 END
```

END OF LISTING

RUN

```
**** FORMALLY CORRECT PROGRAM ****
INTRODUCI I VALORI PER A      ?
1,2,3,4,5,6,7,8,9,1,2,3,4,5,6,7
INTRODUCI I VALORI DI B      ?
10,20,30,40,50,60,70,80,90
INTRODUCI I VALORI DI C      ?
-1,-2,-3,-4,-5,-6,-7,-8,-9,-1,-2,-3,-4,-5,-6,-7,-8,-9,-1,-2
```

I VALORI DI A,B e C, STAMPATI USANDO ; COME SEPARATORE SONO:

```
 1  2  3  4
 5  6  7  8
 9  1  2  3
 4  5  6  7
10 20 30
40 50 60
70 80 90
-1          -2          -3          -4
-5          -6          -7          -8
-9          -1         -2          -3
-4          -5         -6          -7
-8          -9         -1          -2
```

CON DUE ISTRUZIONI PRODUCO LA STAMPA DI PRIMA.

1	2	3	4				
5	6	7	8				
9	1	2	3				
4	5	6	7				
10	20	30					
40	50	60					
70	80	90					
-1		-2		-3			-4
-5		-6		-7			-8
-9		-1		-2			-3
-4		-5		-6			-7
-8		-9		-1			-2

STAMPO I VALORI DI A,B e C USANDO , COME SEPARATORE.

1	2	3	4				
5	6	7	8				
9	1	2	3				
4	5	6	7				
10	20	30					
40	50	60					
70	80	90					
-1		-2		-3			-4
-5		-6		-7			-8
-9		-1		-2			-3
-4		-5		-6			-7
-8		-9		-1			-2

CON DUE ISTRUZIONI PRODUCO LA STAMPA DI PRIMA.

1	2	3	4				
5	6	7	8				
9	1	2	3				
4	5	6	7				
10	20	30					
40	50	60					
70	80	90					
-1		-2		-3			-4
-5		-6		-7			-8
-9		-1		-2			-3
-4		-5		-6			-7
-8		-9		-1			-2

Con due istruzioni produco la stampa di prima

1	2	3	4	5
6	7	8		
9	0	1	2	3
4	5	6		
7	8	9	0	1
2	3	4		
5	6	7	8	9
0	1	2		
3	4	5	6	7
8	9	0		
1	2	3	4	5
6	7	8		
9	0	1	2	3
4	5	6		
7	8	9	0	1
2	3	4		
10	20	30	40	50
60				
70	80	90	10	20
30				
40	50	60	70	80
90				
10	20	30	40	50
60				
70	80	90	10	20
30				
40	50	60	70	80
90				
-1	-2	-3	-4	
-5	-6	-7	-8	
-9	-1	-2	-3	
-4	-5	-6	-7	
-8	-9	-1	-2	

2. La routine sottostante, dopo aver assegnato le stringhe introdotte da tastiera alle variabile multiple stringa A\$ e B\$, le stampa con due formati diversi secondo quanto specificato nelle istruzioni 90 e 150.

```
LIST
FILE
```

```
0010 DISP "Introduci i valori per A$          ";
0020 MAT INPUT A$(2,3)
0030 DISP "Introduci i valori per B$          ";
0040 MAT INPUT B$(2,2)
0050 PRINT
0060 PRINT
0070 PRINT
0080 PRINT "Stampo i valori di A$ e B$ usando ;"
0090 MAT PRINT A$;B$;
0100 PRINT
0110 PRINT
0120 PRINT
0130 PRINT "Stampo i valori di A$ e B$ usando ,"
0140 MAT PRINT A$,B$
0150 END
```

```
END OF LISTING
```

```
RUN
```

```
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per A$          ?
*BASIC P6066*,*BASIC P6066*,*BASIC P6066*,*BASIC P6066*,*BASIC P6066*
Introduci i valori per A$          ??
*BASIC P6066*
Introduci i valori per B$          ?
*Olivetti P6066*,*Olivetti P6066*,*Olivetti P6066*,*Olivetti P6066*
```

```
Stampo i valori di A$ e B$ usando ;
*BASIC P6066**BASIC P6066**BASIC P6066*
*BASIC P6066**BASIC P6066**BASIC P6066*
*Olivetti P6066**Olivetti P6066*
*Olivetti P6066**Olivetti P6066*
```

```
Stampo i valori di A$ e B$ usando ,
*BASIC P6066*  *BASIC P6066*  *BASIC P6066*
*BASIC P6066*  *BASIC P6066*  *BASIC P6066*
*Olivetti P6066*  *Olivetti P6066*
*Olivetti P6066*  *Olivetti P6066*
```

Istruzione MAT PRINT USING

Funzione

Stampa i valori degli elementi di una o più variabili multiple in un formato predefinito in una istruzione immagine.

Formato

MAT PRINT USING {line-num | string-var}, array [, array] ...

dove:

line-num

indica il numero di linea di una istruzione IMMAGINE

string-var

indica una variabile stringa, semplice o con indice, il cui contenuto rappresenta una immagine di formato

array

indica il nome di una variabile multipla, numerica o stringa, i cui valori sono stampati secondo un formato definito dall'utente.

Azione

I valori degli elementi di ogni riga delle matrici specificate nella istruzione MAT PRINT sono convertiti nel formato specificato nella istruzione IMMAGINE il cui numero di linea è specificato con line-num o dal contenuto della variabile stringa, specificata con string-var, e stampati, da sinistra a destra, nell'ordine con cui sono presenti in ogni riga della variabile multipla.

L'associazione tra valori da stampare e campi della immagine di formato è data, da sinistra a destra, nell'ordine con cui i valori compaiono negli elementi della riga della matrice ed "i campi immagine" nella immagine di formato.

Note

1. Ogni istruzione PRINT USING stampa i valori degli elementi di ogni riga delle variabili multiple a partire da una nuova riga di stampa.
2. Se vi sono più elementi nella riga di una variabile multipla che campi di formato, nella immagine di formato, gli elementi in più sono stampati sulle righe di stampa successive con lo stesso formato.
3. Se vi sono più campi di formato nella immagine di formato che elementi in una riga di una variabile multipla, in corrispondenza dei campi immagine eccedenti vengono generati degli spazi.
4. I valori degli elementi di una variabile multipla ed i campi di formato, nella immagine di formato, devono essere coerenti: ad una matrice numerica deve corrispondere una immagine di formato di tipo numerico; ad una variabile multipla stringa deve corrispondere una immagine di formato di tipo stringa.

Esempi

1. Nella routine sottostante l'istruzione 90 stampa i valori delle matrici A e B secondo il formato definito con l'istruzione 50. L'istruzione 150 stampa gli stessi valori con il formato definito dal contenuto della variabile I\$. Poichè il formato è composto da due soli campi, il terzo valore di ogni riga delle matrici A e B è stampato su di una nuova riga di stampa. Infine l'istruzione 210 stampa gli stessi valori con il formato definito dall'istruzione 160. In questo caso, il campo immagine in più è ignorato.

```
LIST
FILE +MAT40
```

```
0010 DISP "Introduci i valori per A      ";
0020 MAT INPUT A(3,3)
0030 DISP "Introduci i valori per B      ";
0040 MAT INPUT B(3,3)
0050 :   ###           ###           ###
0060 PRINT
0070 PRINT
0080 PRINT "Stampo i valori con l'immagine definita nell'istruzione 50"
0090 MAT PRINT USING 50,A,B
0100 LET I$="###           ###"
0110 PRINT
0120 PRINT
0130 PRINT
0140 PRINT "Stampo i valori di A e B con l'immagine contenuta in I$."
0150 MAT PRINT USING I$,A,B
0160 :   ####           ####           ####           ####
0170 PRINT
0180 PRINT
0190 PRINT
0200 PRINT "Stampo i valori con l'immagine definita nell'istruzione 160."
0210 MAT PRINT USING 160,A,B
0220 END
```

END OF LISTING

```
RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per A      ?
1,2,3,4,5,6,7,8,9
Introduci i valori per B      ?
-1,-2,-3,-4,-5,-6,-7,-8,-9
```

```
Stampo i valori con l'immagine definita nell'istruzione 50
      1           2           3
      4           5           6
      7           8           9
     -1          -2          -3
     -4          -5          -6
     -7          -8          -9
```

```
Stampo i valori di A e B con l'immagine contenuta in I$.
      1           2
      3
      4           5
      6
      7           8
      9
     -1          -2
     -3
     -4          -5
     -6
     -7          -8
     -9
```

```
Stampo i valori con l'immagine definita nell'istruzione 160.
      1           2           3
      4           5           6
      7           8           9
     -1          -2          -3
     -4          -5          -6
     -7          -8          -9
```

2. Nella routine sottostante l'istruzione 90 stampa le stringhe assegnate da tastiera ad A\$ e B\$ con il formato definito nell'istruzione 50. L'istruzione 150 stampa le stesse stringhe con il formato definito dal contenuto di I\$. Poichè in questo caso il formato è composto da due soli campi immagine, il terzo valore di ogni riga delle matrici viene stampato su di una nuova riga di stampa. Infine l'istruzione 210 stampa gli stessi valori con il formato definito dall'istruzione 160. In questo caso il formato ha un campo immagine in più del numero di elementi di ogni riga delle variabili multiple stringa per cui tale campo è ignorato.

```

LIST
FILE      +MAT19

0005 DCL 80I$
0010 DISP "Introduci i valori per A$
0020 MAT INPUT A$(3,3)
0030 DISP "Introduci i valori per B$
0040 MAT INPUT B$(3,3)
0050      'LLLLLLLLLLLLLLLL      'CCCCCCCCCCCCCCCC      'RRRRRRRRRRRRRRRR
0060 PRINT
0070 PRINT
0080 PRINT "Stampo i valori con l'immagine definita nell'istruzione 50"
0090 MAT PRINT USING 50,A$,B$
0100 LET I$="LLLLLLLLLLLLLLLL      'LLLLLLLLLLLLLLLL"
0110 PRINT
0120 PRINT
0130 PRINT
0140 PRINT "Stampo i valori di A$ e B$ con l'immagine contenuta in I$."
0150 MAT PRINT USING I$,A$,B$
0160 'LLLLLLLLLLLLLLLL      'LLLLLLLLLLLLLLLL      'LLLLLLLLLLLLLLLL      'LLLLLLLLLLLLLLLL
0170 PRINT
0180 PRINT
0190 PRINT
0200 PRINT "Stampo i valori con l'immagine definita nell'istruzione 160."
0210 MAT PRINT USING 160,A$,B$
0220 END

END OF LISTING

RUN
**** FORMALLY CORRECT PROGRAM ****
Introduci i valori per A$      ?
AAAAAAAAAAAAAAAA,BBBBBBBBBBBBBBBB,CCCCCCCCCCCCCCCC,DDDDDDDDDDDDDDDD
Introduci i valori per A$      ??
EEEEEEEEEEEEEEEE,FFFFFFFFFFFFFF,GGGGGGGGGGGGGGGG,HHHHHHHHHHHHHHHH
Introduci i valori per A$      ??
IIIIIIIIIIIIIIII
Introduci i valori per B$      ?
a,b,c,d,e,f,g,h,i

Stampo i valori con l'immagine definita nell'istruzione 50
AAAAAAAAAAAAAAAA      BBBBBBBBBBBBBBBBB      CCCCCCCCCCCCCCCC
DDDDDDDDDDDDDDDD      EEEEEEEEEEEEEEEE      FFFFFFFFFFFFFFFF
GGGGGGGGGGGGGGGG      HHHHHHHHHHHHHHHH      IIIIIIIIIIIIIIII
a                      b                      c
b                      e                      f
a                      h                      i

```

Stampo i valori di A\$ e B\$ con l'immagine contenuta in I\$.

AAAAAAAAAAAAAAAA	BBBBBBBBBBBBBBBB
CCCCCCCCCCCCCCCC	
DDDDDDDDDDDDDDDD	EEEEEEEEEEEEEEEE
FFFFFFFFFFFFFFFF	
GGGGGGGGGGGGGGGG	HHHHHHHHHHHHHHHH
IIIIIIIIIIIIIIII	
a	b
c	
d	e
f	
g	h
i	

Stampo i valori con l'immagine definite nell'istruzione 150.

AAAAAAAAAAAAAAAA	BBBBBBBBBBBBBBBB	CCCCCCCCCCCCCCCC
DDDDDDDDDDDDDDDD	EEEEEEEEEEEEEEEE	FFFFFFFFFFFFFFFF
GGGGGGGGGGGGGGGG	HHHHHHHHHHHHHHHH	IIIIIIIIIIIIIIII
a	b	c
d	e	f
g	h	i

Istruzione MAT READ

Funzione Assegna agli elementi di una o più variabili multiple i dati contenuti nel file interno definito con le istruzioni DATA.

Formato **MAT READ array [(num-exp, num-exp) [, array [(num-exp, num-exp)]] ...**

dove:

array

indica il nome di una variabile multipla, numerica o stringa, ai cui elementi sono assegnati i valori dal file dati interno

num-exp

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta la nuova dimensione attuale della variabile multipla.

Azione

Agli elementi delle variabili multiple specificate nella istruzione sono assegnati ordinatamente, riga per riga, e variabile per variabile, i valori del file dati interno iniziando dalla posizione indicata dal pointer (vedi istruzione DATA).

Man mano che ogni valore è assegnato ad ogni elemento della variabile multipla, il pointer indica la posizione del dato successivo del file dati interno.

Se sono specificate le espressioni numeriche della parte opzionale, queste ultime sono eseguite ed i valori ottenuti, arrotondati all'intero più prossimo, rappresentano le nuove dimensioni attuali della variabile multipla.

Se non sono specificate le espressioni numeriche suddette, i valori del file dati interno sono assegnati agli elementi della variabile multipla secondo le sue dimensioni attuali.

1. Ogni elemento di una variabile multipla presente in una istruzione MAT READ deve avere un valore associato nel file dati interno, altrimenti l'esecuzione del programma è sospesa e viene visualizzato un messaggio di errore non recuperabile. Premendo **BREAK** il sistema commuta nello stato comandi.
2. Agli elementi delle variabili multiple di tipo stringa possono essere assegnati dati numerici che sono considerati stringhe di caratteri.
3. Ad un elemento di una matrice numerica non deve essere assegnata una stringa.
4. Se ad un elemento di una matrice dichiarata in singola precisione è assegnato un dato numerico in virgola mobile con più di 6 cifre significative, ma con esponente che rientri nel range della singola precisione, la mantissa viene troncata a 6 cifre significative.
5. Se ad un elemento di una matrice dichiarata in singola precisione è assegnato un dato numerico in virgola mobile con esponente nella zona di OVERFLOW per la singola precisione, il sistema visualizza un messaggio di errore recuperabile e commuta nello stato di debugging, assegnando all'elemento suddetto il valore 9.99999E63 oppure -9.99999E63. Premendo **CONTINUE** l'elaborazione del programma prosegue; premendo **BREAK** l'elaborazione termina.
6. Se ad un elemento di una matrice dichiarata in singola precisione è assegnato un dato numerico in virgola mobile con esponente nella zona di UNDERFLOW per il tipo di precisione specificato, il sistema visualizza un messaggio di errore recuperabile e commuta nello stato di debugging, assegnando all'elemento suddetto il valore zero. Premendo **CONTINUE** l'elaborazione del programma prosegue; premendo **BREAK** l'elaborazione termina.
7. Se ad un elemento di una variabile multipla di tipo stringa viene assegnata una stringa con più caratteri della lunghezza di allocazione dichiarata per gli elementi della variabile multipla, il sistema visualizza un messaggio di errore recuperabile e commuta nello stato di debugging, assegnando all'elemento suddetto la stringa troncata sulla destra

dei caratteri eccedenti. Premendo **CONTINUE** l'elaborazione del programma prosegue; premendo **BREAK** l'elaborazione termina.

Esempio

Nella routine sottostante tutte le variabili numeriche sono rappresentate in singola precisione e tutti gli elementi di A\$ possono avere al massimo 30 caratteri. Le istruzioni 30, 40 e 50 definiscono un file dati interno da cui sono prelevati i valori mediante le istruzioni MAT READ. L'istruzione 60 assegna ai primi due elementi delle prime due righe della matrice A i primi quattro valori del file dati interno suddetto. Si noti come per il primo ed ultimo valore la mantissa è stata troncata dopo le prime 6 cifre significative, perchè la matrice ha elementi in singola precisione. I successivi quattro valori del file dati interno sono assegnati agli elementi di A\$ con l'istruzione 160. L'istruzione 180 ripone il pointer del file dati interno all'inizio del file. L'istruzione 220, infine, assegna ai primi due elementi delle prime due righe di A\$ (le dimensioni attuali sono state definite con la precedente istruzione 160) i primi 4 valori contenuti nel file dati interno, come si vede dalla stampa prodotta con l'istruzione 240.

```
FILE      RMATRD

0010 DCL SINGLE
0020 DCL 30(A$( ))
0030 DATA 123456789E34,123456E34,-123456E-34,-123456789E-34
0040 DATA * Olivetti P6066*,"AREA,VOLUME"," PESO  "
0050 DATA Temperatura "media" a Parigi
0060 MAT READ A(2,2)
0070 PRINT
0080 PRINT
0090 PRINT
0100 PRINT "I valori di A sono:"
0110 MAT PRINT A;
0120 PRINT
0130 PRINT
0140 PRINT
0150 PRINT "I valori di A$ sono:"
0160 MAT READ A$(2,2)
0170 MAT PRINT A$;
0180 RESTORE
0190 PRINT
0200 PRINT
0210 PRINT
0220 MAT READ A$
0230 PRINT "Ora i valori di A$ sono:"
0240 MAT PRINT A$
0250 END

END OF LISTING
```

I valori di A sono:
1.2345600E+42 1.2345600E+39
-1.2345600E-29 -1.2345600E-26

I valori di A\$ sono:
* Olivetti P6066*AREA,UOLUME
PESO Temperatura "media" a Parigi

Ora i valori di A\$ sono:
123456789E34 123456E34
-123456E-34 -123456789E-34

Istruzione MAT READ:

Funzione Assegna agli elementi di una o più variabili multiple i dati contenuti in un file dati esterno.

Formato **MAT READ: file-designator, array [(num-exp, num-exp)] [, array [(num-exp, num-exp)] ... [EOF line-num]**

dove:

file-designator

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta il designatore di un file dati esterno

array

indica il nome di una variabile multipla numerica o stringa

num-exp

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, rappresenta la nuova dimensione attuale della variabile multipla

line-num

è il numero di linea di una istruzione del programma.

Azione

L'espressione numerica relativa a file designator è eseguita ed il valore ottenuto, arrotondato all'intero più prossimo nd, specifica il file dati esterno da cui sono prelevati i valori da assegnare agli elementi delle variabili multiple indicate nella istruzione.

I valori da assegnare sono prelevati dal file con numero designatore nd, iniziando dal dato su cui è posizionato il pointer del file, ed assegnati nell'ordine, riga per riga, variabile multipla per variabile multipla, agli elementi delle variabili multiple presenti nella istruzione.

Il pointer del file si sposta man mano e si posiziona dopo l'ultimo dato letto.

Se una variabile multipla è seguita dalla parte opzionale, le espressioni numeriche sono eseguite ed i valori ottenuti, arrotondati all'intero più prossimo m ed n , specificano quali sono le nuove dimensioni attuali della matrice. Quindi i valori del file sono assegnati ad una matrice di dimensioni attuali $m*n$.

Se dopo una variabile multipla non si specifica la parte opzionale, i valori del file dati esterno sono assegnati alla variabile multipla suddetta secondo le sue dimensioni attuali.

Se il pointer del file è posizionato dopo l'ultimo dato registrato in un file sequenziale, l'esecuzione dell'istruzione MAT READ: dà una segnalazione di errore e l'esecuzione del programma è sospesa; se, però, è presente l'opzione EOF, il controllo della esecuzione del programma passa alla istruzione il cui numero di linea è specificato nella opzione stessa e non vi è alcuna segnalazione di errore.

Se il pointer di un file ad accesso diretto è posizionato dopo l'ultima parola allocata per il file (vedi comando CREATE), l'esecuzione della istruzione MAT READ: dà una segnalazione di errore; se, però, è presente l'opzione EOF, il controllo della esecuzione del programma passa alla istruzione il cui numero di linea è specificato nella opzione stessa e non vi è alcuna segnalazione di errore.

Note

1. Se il file è stato dichiarato, con il comando CREATE, ad accesso diretto, si possono leggere i dati che si vogliono. Infatti l'istruzione SETW: posiziona il pointer del file su un dato qualunque del file esterno.
2. Se il file è stato dichiarato, con il comando CREATE, ad accesso sequenziale, l'istruzione MAT READ: deve essere preceduta da una delle seguenti istruzioni riferite allo stesso file: RESTORE:, READ:, MAT READ:.
3. I valori assegnati agli elementi delle variabili multiple devono essere dello stesso tipo di queste

ultime (numeriche o stringa).

4. Il risultato della espressione numerica, arrotondato all'intero più prossimo, che determina il numero designatore del file esterno da cui sono letti i valori deve essere maggiore di zero e minore od uguale al numero di file accessibili contemporaneamente dal programma (dichiarato con l'istruzione FILES).
5. Il file esterno deve essere stato "aperto" mediante una precedente istruzione FILES o FILE:.
6. Per ulteriori osservazioni si vedano le note contenute nella descrizione dell'istruzione READ:.

Esempi

1. La routine sottostante è eseguita dopo aver eseguito la routine riportata nell'esempio 1 della descrizione della istruzione WRITE: . L'istruzione 10 assegna ai file sequenziali SEQ1 e SEQ2 i numeri designatori 1 e 2. L'istruzione 20 pone il file SEQ1 in lettura ed il suo pointer all'inizio del file. L'istruzione 30 legge i dati del file esterno SEQ1 e li assegna ai primi tre elementi delle prime tre righe di A ed ai primi tre elementi delle prime tre righe di B. L'istruzione 80 stampa i valori assegnati ad A da file dati esterno. L'istruzione 130 stampa i valori assegnati a B da file dati esterno. L'istruzione 140 pone il pointer del file SEQ2 all'inizio del file e ne permette la lettura. L'istruzione 150 legge le stringhe di dati del file SEQ2 e le assegna ai primi due elementi delle prime due righe della variabile multipla A\$ ed ai primi tre elementi delle prime due righe di B\$. Le istruzioni 190 e 240 stampano i valori assegnati ad A\$ e B\$.

```
LIST
FILE      MATRE1

0010 FILES SEQ1;SEQ2
0020 RESTORE :1
0030 MAT READ :1,A(3,3),B(2,3)
0040 PRINT
0050 PRINT
0060 PRINT
```

```

0070 PRINT "I valori di A() sono:"
0080 MAT PRINT A;
0090 PRINT
0100 PRINT
0110 PRINT
0120 PRINT "I valori di B() sono:"
0130 MAT PRINT B
0140 RESTORE :2
0150 MAT READ :2,A$(2,2),B$(2,3)
0160 PRINT
0170 PRINT
0180 PRINT
0190 PRINT "I valori di A$ sono:"
0200 MAT PRINT A$
0210 PRINT
0220 PRINT
0230 PRINT
0240 PRINT "I valori di B$ sono:"
0250 MAT PRINT B$
0255 END

```

END OF LISTING

```

RUN
**** FORMALLY CORRECT PROGRAM ****

```

```

I valori di A() sono:
 1  2  3
 4  5  6
 7  8  9

```

```

I valori di B() sono:
 2           4           6
 8           16          25

```

```

I valori di A$ sono:
PRIMO DATO   SECONDO DATO
TERZO DATO   QUARTO DATO

```

```

I valori di B$ sono:
Biella       Milano       Napoli
Roma         Torino       Venezia

```

2. La routine sottostante è eseguita dopo aver eseguito la routine riportata nell'esempio 1 della descrizione dell'istruzione WRITE:. L'istruzione 10 assegna ai file esterni, ad accesso diretto, DIR1 e DIR2 i numeri designatori 1 e 2. L'istruzione 20 pone il pointer del file DIR1 all'inizio della decima parola. L'istruzione 30 legge dalla decima parola del file DIR1 i dati da assegnare ai primi tre elementi delle prime tre righe di A. L'istruzione 40 pone il pointer del file DIR1 all'inizio della 30-esima parola e quindi l'istruzione 50 assegna i dati che iniziano da tale parola ai primi due elementi delle prime due righe di B. L'istruzione 60 legge dall'inizio del file DIR2 i dati e

li assegna ordinatamente ai primi due elementi delle prime due righe della variabile A\$ ed ai primi tre elementi delle prime due righe della variabile B\$. Le istruzioni 110, 160, 210 e 260 stampano i valori assegnati alle matrici A e B ed alle variabili multiple stringa A\$ e B\$.

```
LIST
FILE      +MATRE2

0010 FILES DIR1;DIR2
0020 SETW :1 TO 10
0030 MAT READ :1,A(3,3)
0040 SETW :1 TO 30
0050 MAT READ :1,B(2,2)
0060 MAT READ :2,A$(2,2),B$(2,3)
0070 PRINT
0080 PRINT
0090 PRINT
0100 PRINT "I valori di A() sono:"
0110 MAT PRINT A;
0120 PRINT
0130 PRINT
0140 PRINT
0150 PRINT "I valori di B() sono:"
0160 MAT PRINT B;
0170 PRINT
0180 PRINT
0190 PRINT
0200 PRINT " I valori di A$() sono:"
0210 MAT PRINT A$
0220 PRINT
0230 PRINT
0240 PRINT
0250 PRINT "I valori di B$() sono:"
0260 MAT PRINT B$
0270 END
```

END OF LISTING

RUN

```
I valori di A() sono:
 1  2  3
 4  5  6
 7  8  9
```

```
I valori di B() sono:
 2  4
 5  8
```

```
I valori di A$() sono:
PRIMO DATO      SECONDO DATO
TERZO DATO      QUARTO DATO
```

```
I valori di B$() sono:
Biella          Milano      Napoli
Roma            Torino      Venezia
```


Istruzione MAT WRITE:

Funzione

Registra in un file dati esterno i valori degli elementi di una o più variabili multiple specificate.

Formato

MAT WRITE: file-designator, array [, array] ... [EOF line-num]

dove:

file-designator

è una espressione numerica il cui valore, arrotondato all'intero più prossimo, indica il designatore del file dati esterno in cui devono essere registrati i valori degli elementi delle variabili multiple specificate

array

è il nome di una variabile multipla da cui sono prelevati i valori da registrare sul file dati esterno

line-num

indica il numero di linea di una istruzione del programma.

Azione

L'espressione riferita a file-designator è eseguita ed il valore ottenuto, arrotondato all'intero più prossimo nd, costituisce il numero designatore del file nel quale saranno registrati i valori degli elementi delle variabili multiple specificate nella istruzione.

I valori degli elementi delle variabili multiple indicate nella istruzione sono registrati nel file suddetto, iniziando dalla posizione indicata dal pointer del file. La registrazione avviene elemento dopo elemento, riga dopo riga, nell'ordine da sinistra a destra per ogni variabile multipla indicata nell'istruzione. Se il pointer del file è posizionato dopo l'ultima parola allocata per il file (sequenziale o ad accesso diretto)

nella libreria, l'esecuzione dell'istruzione MAT WRITE: dà una segnalazione di errore non recuperabile; se, però, è presente l'opzione EOF, il controllo della esecuzione del programma passa alla istruzione il cui numero di linea è specificato nella opzione stessa con line-num e non vi è alcuna segnalazione di errore. In ogni caso i valori per i quali esiste sufficiente spazio sono registrati nel file.

Note

1. Se il file è sequenziale, l'istruzione MAT WRITE: deve essere preceduta da una istruzione SCRATCH, APPEND:, WRITE:, o MAT WRITE:.
2. Se il file è ad accesso casuale e si vogliono registrare i dati iniziando da una parola specifica del file, l'istruzione MAT WRITE: deve essere preceduta da una istruzione SETW:.
3. Se l'istruzione MAT WRITE: è eseguita dopo una istruzione FILES, FILE:, SCRATCH: o RESTORE: la registrazione del file inizia dalla prima parola del file stesso (l'istruzione SCRATCH: è usata solo per file sequenziali).
4. nd deve essere maggiore di zero e minore o uguale al numero di file accessibili contemporaneamente dal programma (dichiarato con l'istruzione FILES).
5. Per ulteriori informazioni si vedano le note riportate nella descrizione della istruzione WRITE:.

Esempi

1. Nella routine sottostante l'istruzione 10 assegna ai file SEQ1 e SEQ2 (sequenziali) e DIR1 e DIR2 (ad accesso diretto) rispettivamente i numeri designatori 1,2,3 e 4. L'istruzione 100 pone il pointer del file SEQ1 nella prima posizione del file e permette di registrare in esso i valori contenuti nelle matrici A e B (istruzione 110). L'istruzione 120 pone il pointer del file SEQ2 nella prima posizione del file e permette di registrare in esso i valori delle variabili multiple A\$ e B\$ con l'istruzione 130. L'istruzione 140 pone il pointer del file DIR1 all'inizio della decima parola. L'istruzione 150 registra dalla decima parola del file DIR1 i valori della matrice A. L'istruzione 160 pone il pointer del file DIR1 all'inizio della 30-esima parola da

dove l'istruzione 170 registra i valori della matrice B. Infine l'istruzione 180 registra dall'inizio del file DIR2 le stringhe contenute negli elementi delle variabili multiple A\$ e B\$.

```

LIST
FILE      MATWR1

0010 FILES SEQ1;SEQ2;DIR1;DIR2
0020 DISP "Introduci i valori per A()      "
0030 MAT INPUT A(3,3)
0040 DISP "Introduci i valori per B()      "
0050 MAT INPUT B(2,3)
0060 DISP "Introduci i valori per A$( )    "
0070 MAT INPUT A$(2,2)
0080 DISP "Introduci i valori per B$( )    "
0090 MAT INPUT B$(2,3)
0100 SCRATCH :1
0110 MAT WRITE :1,A,B
0120 SCRATCH :2
0130 MAT WRITE :2,A$,B$
0140 SETW :3 TO 10
0150 MAT WRITE :3,A
0160 SETW :3 TO 30
0170 MAT WRITE :3,B
0180 MAT WRITE :4,A$,B$
0190 END

END OF LISTING

```

```

RUN
Introduci i valori per A()      ?
1,2,3,4,5,6,7,8,9
Introduci i valori per B()      ?
2,4,6,8,16,25
Introduci i valori per A$( )    ?
PRIMO DATO,SECONDO DATO,TERZO DATO,QUARTO DATO
Introduci i valori per B$( )    ?
Biella,Milano,Napoli,Roma,Torino,Venezia

```

2. Nella routine sottostante l'istruzione 10 assegna ai file ad accesso diretto DIRE1 e DIRE2 i numeri designatori 1 e 2. L'istruzione 40 pone il pointer del file DIRE1 all'inizio della decima parola del file da dove i valori della matrice A sono registrati con l'istruzione 50. L'istruzione 80 pone il pointer del file DIRE2 all'inizio della 20-esima parola da dove l'istruzione 90 registra i valori degli elementi della matrice B. L'istruzione 100 pone il pointer del file DIRE1 all'inizio della decima parola del file, da dove l'istruzione 110 legge i valori da assegnare ai primi 5 elementi delle prime 5 righe della matrice A. L'istruzione 170 pone il pointer del file DIRE2 all'inizio della 20-esima parola da dove l'istruzione 180 legge i valori da assegnare ai primi 5 elementi delle prime 5 ri-

ghe di B. L'istruzione 200 assegna a C il prodotto, righe per colonne, delle matrici A e B. L'istruzione 210 pone il pointer del file DIRE1 all'inizio della decima parola da dove l'istruzione 220 registra i valori della matrice C, aggiornando così il file dati. Infine l'istruzione 230 ripone il pointer del file DIRE1 all'inizio della decima parola da dove l'istruzione 240 legge i valori da assegnare nuovamente ai primi 5 elementi delle prime 5 righe di C.

```
LIST
FILE      MATRU
```

```
0010 FILES DIRE1;DIRE2
0020 DISP "Introduzione dei valori per A()  "
0030 MAT INPUT A(5,5)
0040 SETW :1 TO 10
0050 MAT WRITE :1,A
0060 DISP "Introduci i valori per B()      "
0070 MAT INPUT B(5,5)
0080 SETW :2 TO 20
0090 MAT WRITE :2,B
0100 SETW :1 TO 10
0110 MAT READ  :1,A
0120 PRINT
0130 PRINT
0140 PRINT
0150 PRINT "I valori di A() sono:"
0160 MAT PRINT A;
0170 SETW :2 TO 20
0180 MAT READ  :2,B(5,5)
0181 PRINT
0182 PRINT
0183 PRINT
0184 PRINT "I valori di B() sono:"
0190 MAT PRINT B;
0200 MAT C=A*B
0210 SETW :1 TO 10
0220 MAT WRITE :1,C
0230 SETW :1 TO 10
0240 MAT READ  :1,C(5,5)
0250 PRINT
0260 PRINT
0270 PRINT
0280 PRINT "I valori di registrati nel file DIRE1 a partire dalla 10a parola : "
0290 MAT PRINT C
0300 END
```

END OF LISTING

RUN

```
**** FORMALLY CORRECT PROGRAM ****
Introduzione dei valori per A()  ?
1,2,3,4,5,6,7,8,9,1,2,3,4,5,6,7,8,9,1,2,3,4,5,6,7
Introduci i valori per B()      ?
1,2,3,4,5,6,7,8,9,1,2,3,4,5,6,7,8,9,1,2,3,4,5,6,7
```

I valori di A() sono:

```
1 2 3 4 5
6 7 8 9 1
2 3 4 5 6
7 8 9 1 2
3 4 5 6 7
```

I valori di B() sono:

1	2	3	4	5
6	7	8	9	1
2	3	4	5	6
7	8	9	1	2
3	4	5	6	7

I valori di registrati nel file DIRE1 a partire dalla 10a parola :

62	77	92	71	68
130	161	192	142	110
81	101	121	96	89
86	113	140	158	113
100	125	150	121	110

